

# software

## TI-99/4A

### Viaggio nello spazio

di Marco Limbiati - Angera (VA)

Il gioco consiste nell'immaginaria esplorazione di un'ampia zona dello spazio siderale alla ricerca di pianeti abitati. Lo schermo e la tastiera del calcolatore fungeranno da pannello di comando della nostra astronave attraverso il quale riceveremo le informazioni dallo spazio circostante e controlleremo la nostra rotta. Il gioco non è grafico, per cui non ci saranno mappe stellari, disegni di pianeti e stelle, basi nemiche o astronavi; tutto ciò viene rappresentato in forma numerica e tramite messaggi che compaiono sullo schermo, come in definitiva potrebbe accadere sul pannello strumenti di una nave spaziale che non sia proprio "dell'ultima generazione".

Per rappresentare lo spazio e quanto in esso contenuto viene fissato un sistema di assi cartesiani, ossia di tre rette che si incro-

ciano nel punto di coordinate 0,0,0; ognuna di queste tre rette forma un angolo di 90 gradi con le altre, e insieme individuano le tre direzioni che ci permettono di identificare ogni singolo punto dello spazio. La direzione destra - sinistra viene identificata con l'asse X, quella alto - basso con l'asse Y, quella avanti-indietro con l'asse Z. Lo spazio preso in considerazione nel programma è un cubo avente lo spigolo lungo 16.000, avente un vertice nell'origine e situato nel primo ottante. I valori utili per le coordinate vanno quindi da 0 a 16.000.

All'inizio del programma vengono generate in maniera casuale 50 terne di valori, rappresentanti le coordinate di 50 oggetti spaziali; tali valori vengono posti in tre matrici e in base all'indice viene stabilito il tipo di oggetto cui le coordinate si riferiscono. Poiché la disposizione degli oggetti avviene in maniera casuale, la loro distri-

buzione non è perfettamente omogenea, e può capitare di trovare zone di spazio piuttosto vuote e altre molto "affollate".

All'inizio del gioco la nave si trova nel punto di coordinate 0,0,0 e li resta finché noi non interveniamo agendo sui comandi. I tasti guida sono Q,W,E,A,D,S. I primi tre servono per imprimere una accelerazione positiva di valore 100 nella direzione di uno dei tre assi coordinati, rispettivamente X, Y e Z. I secondi tre svolgono una funzione opposta producendo una accelerazione negativa di valore -100 sempre nelle direzioni degli assi coordinati, facendo quindi rallentare la nave o anche retrocedere.

Una volta impressa una spinta alla nave, secondo le leggi dell'inerzia, se non intervengono forze esterne questa prende una velocità che si mantiene costante nel tempo. Se diamo più spinte in direzioni diver-

```

100 CALL CLEAR
110 RANDOMIZE
120 OPTION BASE 1
130 DIM A(50),B(50),E(50)
140 FOR I=1 TO 50
150 A(I)=INT(RND*14000)+1000
160 B(I)=INT(RND*14000)+1000
170 E(I)=INT(RND*14000)+1000
180 NEXT I
190 DEF D(I)=INT((A(I)-X)^2+(B(I)-Y)^2+(E(I)-Z)^2)^(1/2)
200 PRINT "STAI PER INIZIARE IL TUO"
210 PRINT "VIAGGIO NELLA GALASSIA"
220 PRINT "IN CERCA DI PIANETI ABITATI."
230 PRINT "BUONA FORTUNA! :*****"
240 FOR H=1 TO 1000
250 NEXT H
260 CALL CLEAR
270 GOSUB 350
280 CALL KEY(0,K,S)
290 IF K=82 THEN 1710
300 GOSUB 680
310 GOSUB 950
320 GOTO 1000
330 GOSUB 1310
340 GOTO 280
345 REM ***MASCHERA DI STAMPA VIDEO***
350 PRINT "POSIZIONE";TAB(20);"TEMPO "
360 PRINT "ASTRONAVE";TAB(16);"VELOCITA' "
370 PRINT "X=";TAB(10);"VX=";TAB(20);"AX="
380 PRINT "Y=";TAB(10);"VY=";TAB(20);"AY="
390 PRINT "Z=";TAB(10);"VZ=";TAB(20);"AZ="
400 PRINT "PER USARE IL RADAR PREMERE R"
410 PRINT "-----"
420 PRINT ":::::"
430 PRINT "-----"
440 PRINT "X=";TAB(14);"XM="
450 PRINT "Y=";TAB(14);"YM="
460 PRINT "Z=";TAB(14);"ZM="
470 PRINT "DISTANZA="
480 PRINT "-----"
490 PRINT ":::::"
500 PRINT "-----"
510 RETURN

520 REM ***STAMPA NUMERI***
530 CALL HCHAR(R,C,32,6)
540 FOR NM=LEN(A$) TO 1 STEP -1
550 CALL HCHAR(R,C+NM,ASC(SEG$(A$,NM,1)))
560 NEXT NM
570 RETURN
580 REM ***STAMPA STRINGHE***
590 R=19
600 C=1
610 CALL HCHAR(R,C,32,30)
620 FOR NM=1 TO LEN(A$)
630 CALL SOUND(20,4000,10)
640 CALL HCHAR(R,C+NM,ASC(SEG$(A$,NM,1)))
650 NEXT NM
660 RETURN
670 REM ***CONTROLLO ASTRONAVE***
680 CALL KEY(0,K,S)
690 IF AV=0 THEN 720
700 IF RND>.85 THEN 4380
710 IF AV=1 THEN 800
720 IF S=D THEN 800
730 CALL SOUND(40,2000,2)
740 IF K=81 THEN 820
750 IF K=65 THEN 840
760 IF K=87 THEN 860
770 IF K=83 THEN 880
780 IF K=69 THEN 900
790 IF K=68 THEN 920
800 VV=1
810 GOTO 930
820 VX=VX+100
830 GOTO 930
840 VX=VX-100
850 GOTO 930
860 VY=VY+100
870 GOTO 930
880 VY=VY-100
890 GOTO 930
900 VZ=VZ+100
910 GOTO 930
920 VZ=VZ-100
930 RETURN
940 REM ***AGGIORNAMENTO COORDINATE NAVE***

```

```

950 X=X+VX
960 Y=Y+VY
970 Z=Z+VZ
980 RETURN
990 REM VERIFICA SE C'E' CAMPO DI
995 REM INFLUENZA DI CORPI CELESTI
1000 CALL SOUND(50,2000,2)
1010 IF RND>.985 THEN 4330
1020 FOR I=50 TO 1 STEP -1
1030 IF ABS(A(I)-X)>1500 THEN 1140
1040 IF ABS(B(I)-Y)>1500 THEN 1140
1050 IF ABS(E(I)-Z)>1500 THEN 1140
1060 DK=D(I)
1070 IF (I<=50)*(I>43)*(DK<500)THEN 3880
1080 IF (I<=47)*(I>43)*(DK<1500)THEN 2840
1090 IF (I<=43)*(I>38)*(DK<1000)THEN 3340
1100 IF (I<=38)*(I>30)*(DK<1000)THEN 2950
1110 IF (I<=30)*(I>20)*(DK<1500)THEN 3880
1120 IF (I<=80)*(I>20)*(DK<1500)THEN 2570
1130 IF (I<=20)*(I>=1)*(DK<1500)THEN 3450
1140 NEXT I
1150 REM *CANCELLA I MESSAGGI QUANDO SI ESCE
1155 REM DALL'INFLUENZA DI UN CORPO CELESTE*
1160 IF (QQ=0)*(RR=0)*(SS=0)*(CC=0)THEN 1280
1170 CALL HCHAR(19,1,32,150)
1180 CALL HCHAR(17,13,32,7)
1190 FOR U=1 TO 3
1200 CALL HCHAR(13+U,5,32,6)
1210 CALL HCHAR(13+U,19,32,6)
1220 CALL HCHAR(2+U,25,32,6)
1230 NEXT U
1240 QQ=0
1250 CC=0
1260 RR=0
1270 SS=0
1280 CALL SOUND(50,2000,2)
1290 GOTO 330
1300 REM ***AGGIORNAMENTO SCHERMO N.1***
1310 T=T+1
1320 R=1
1330 C=28
1340 A$=STR$(T)
1350 GOSUB 530
1360 IF VV=1 THEN 1430
1370 R=2
1380 C=27
1390 V=INT((VX^2+VY^2+VZ^2)^(1/2))
1400 IF V>450 THEN 3870
1410 A$=STR$(V)
1420 GOSUB 530
1430 A$=STR$(X)
1440 R=3
1450 C=5
1460 GOSUB 530
1470 A$=STR$(Y)
1480 R=4
1490 C=5
1500 GOSUB 530
1510 A$=STR$(Z)
1520 R=5
1530 C=5
1540 GOSUB 530
1550 IF VV=1 THEN 1680
1560 A$=STR$(VX)
1570 R=3
1580 C=15
1590 GOSUB 530
1600 A$=STR$(VY)
1610 R=4
1620 C=15
1630 GOSUB 530
1640 A$=STR$(VZ)
1650 R=5
1660 C=15
1670 GOSUB 530
1680 VV=0
1690 RETURN
1700 REM ***RADAR***
1710 R=8
1720 C=1
1730 CALL SOUND(300,3000,2)
1740 CALL HCHAR(8,1,32,160)
1750 IF AV=0 THEN 1810
1760 A$="RADAR GUASTO"
1770 R=10
1780 C=10
1790 GOSUB 610
1800 GOTO 1980
1810 FOR I=50 TO 1 STEP -1
1820 IF ABS(A(I)-X)>3000 THEN 1910
1830 IF ABS(B(I)-Y)>3000 THEN 1910
1840 IF ABS(E(I)-Z)>3000 THEN 1910
1850 IF D(I)>3000 THEN 1910
1860 IF (I<=50)*(I>43)THEN 2000
1870 IF (I<=43)*(I>38)THEN 2020
1880 IF (I<=38)*(I>30)THEN 2040
1890 IF (I<=30)*(I>20)THEN 2060
1900 IF (I<=20)THEN 2080
1910 NEXT I
1920 CALL GCHAR(8,3,CH)
1930 IF CH<>32 THEN 1980
1940 A$="NIENTE DA SEGNALARE"
1950 R=10
1960 C=8
1970 GOSUB 610
1980 CALL SOUND(500,3000,2)
1990 GOTO 300
2000 A$="BUCCO NERO"
2010 GOTO 2090
2020 A$="CAMPO FORZE"
2030 GOTO 2090
2040 A$="BASE NEMICA"
2050 GOTO 2090
2060 A$="STELLA"
2070 GOTO 2090
2080 A$="PIANETA"
2090 C=1
2100 GOSUB 620
2110 C=13
2120 A$=STR$(A(I))
2130 GOSUB 540
2140 C=19
2150 A$=STR$(B(I))
2160 GOSUB 540
2170 C=25
2180 A$=STR$(E(I))
2190 GOSUB 540
2200 R=R+1
2210 GOTO 1910
2220 REM ***AGGIORNAMENTO SCHERMO N.2***
2230 IF CC=I THEN 2370
2240 CC=1
2250 A$=STR$(A(I))
2260 R=14
2270 C=5
2280 GOSUB 530
2290 A$=STR$(B(I))
2300 R=15
2310 C=5
2320 GOSUB 530
2330 A$=STR$(E(I))
2340 R=16
2350 C=5
2360 GOSUB 530
2370 A$=STR$(INT(D(I)))
2380 R=17
2390 C=15
2400 GOSUB 530
2410 RETURN
2420 REM ***AGGIORNAMENTO SCHERMO N.3***
2430 A$=STR$(AX)
2440 R=3
2450 C=25
2460 GOSUB 530
2470 A$=STR$(AY)
2480 R=4
2490 C=25
2500 GOSUB 530
2510 A$=STR$(AZ)
2520 R=5
2530 C=25
2540 GOSUB 530
2550 RETURN
2560 REM ***STELLA***
2570 K=10000000
2580 IF QQ=1 THEN 2650
2590 QQ=1
2600 A$="ATTENZIONE,SEI NEL CAMPO"
2610 GOSUB 590
2620 A$="DI GRAVITA' DI UNA STELLA"
2630 R=20
2640 GOSUB 600
2650 GOSUB 2680
2660 GOTO 280
2670 REM ***CALCOLO ATTRAZIONE DI GRAVITA'***
2680 G=K/(D(I)^3)
2690 AX=INT(G*(A(I)-X))
2700 AY=INT(G*(B(I)-Y))
2710 AZ=INT(G*(E(I)-Z))
2720 VX=VX+AX

```

*(continua a pagina 148)*

(segue da pagina 147)

```
2730 VY=VY+AY
2740 VZ=VZ+AZ
2750 X=X+AX
2760 Y=Y+AY
2770 Z=Z+AZ
2780 VV=0
2790 GOSUB 2230
2800 GOSUB 2430
2810 GOSUB 1310
2820 RETURN
2830 REM ***BUCO NERO***
2840 IF RR=1 THEN 2910
2850 RR=1
2860 A$="ATTENZIONE,SEI NEL CAMPO"
2870 GOSUB 590
2880 A$="DI GRAVITA' DI UN BUCO NERO"
2890 R=20
2900 GOSUB 600
2910 K=50000000
2920 GOSUB 2600
2930 GOTO 280
2940 REM ***BASE NEMICA***
2950 A$="ALLARME,BASE NEMICA"
2960 GOSUB 590
2970 GOSUB 1310
2980 GOSUB 2230
2990 A$="ATTENZIONE,MINA NEMICA"
3000 R=20
3010 GOSUB 600
3020 XM=X+2*VX
3030 YM=Y+2*VY
3040 ZM=Z+2*VZ
3050 A$=STR$(XM)
3060 R=14
3070 C=19
3080 GOSUB 530
3090 A$=STR$(YM)
3100 R=15
3110 C=19
3120 GOSUB 530
3130 A$=STR$(ZM)
3140 R=16
3150 C=19
3160 GOSUB 530
3170 GOSUB 680
3180 GOSUB 950
3190 GOSUB 1310
3200 GOSUB 680
3210 GOSUB 950
3220 GOSUB 1310
3230 DI=((XM-X)^2+(YM-Y)^2+(ZM-Z)^2)^(1/2)
3240 IF DI<30 THEN 3880
3250 FOR O=1 TO 16
3260 CALL SCREEN(0)
3270 NEXT O
3280 CALL SCREEN(4)
3290 A$="MINA EVITATA"
3300 R=20
3310 GOSUB 600
3320 GOTO 280
3330 REM ***CAMPO DI FORZE***
3340 CALL HCHAR(19,1,32,160)
3350 A$="ATTENZIONE,CAMPO DI FORZE"
3360 GOSUB 590
3370 VV=0
3380 VX=INT(VX/3)
3390 VY=INT(VY/3)
3400 VZ=INT(VZ/3)
3410 GOSUB 1310
3420 GOSUB 2230
3430 GOTO 280
3440 REM ***PIANETA***
3450 GOSUB 1310
3460 GOSUB 2230
3470 IF SS=1 THEN 3600
3480 SS=1
3490 A$="SIAMO VICINI A UN PIANETA!"
3500 GOSUB 590
3510 A$="SE TI AVVICINI A DISTANZA"
3520 R=20
3530 GOSUB 600
3540 A$="INFERIORE A 500 UNITA'"
3550 R=21
3560 GOSUB 600
3570 A$="PUOI SCOPRIRE SE E' ABITATO"
3580 R=22
3590 GOSUB 600
3600 IF D(I)<300 THEN 3880
3610 IF D(I)<500 THEN 3630
3620 GOTO 280
3630 IF RND>0.60 THEN 3750
3640 CALL HCHAR(19,1,32,150)
3650 A$="PURTROPP0 NON VIENE RILEVATA"
3660 GOSUB 590
3670 A$="ALCUNA TRACCIA DI VITA"
3680 R=20
3690 GOSUB 600
3700 A(I)=INT(RND*14000)+1000
3710 B(I)=INT(RND*14000)+1000
3720 E(I)=INT(RND*14000)+1000
3730 GOTO 280
3750 IF PP=2 THEN 4030
3760 CALL HCHAR(19,1,32,150)
3770 A$="HOLTO BENE,HAI SCOPERTO"
3780 GOSUB 590
3790 A$="UN PIANETA ABITABILE"
3800 R=20
3810 GOSUB 600
3820 A$="MA DEVI CERCARE ANCORA."
3830 R=21
3840 GOSUB 600
3850 PP=PP+1
3860 GOTO 3700
3865 REM ***FINE GIOCO***
3870 ZZ=1
3880 FOR S=1 TO 15
3890 CALL SCREEN(S)
3900 CALL SOUND(100,110,5*2,-6,5*2)
3910 NEXT S
3920 CALL SCREEN(4)
3930 CALL CLEAR
3940 IF ZZ=1 THEN 3970
3950 PRINT " ASTRONAVE DISTRUTTA !":*****
3960 GOTO 3990
3970 PRINT " VELOCITA' SUPERIORE A 450!"
3975 PRINT " ASTRONAVE ESPL0SA":*****
3980 ZZ=0
3990 FOR S=1 TO 500
4000 NEXT S
4010 CALL CLEAR
4020 GOTO 4150
4030 A$="COMPLIMENTI,HAI PORTATO A"
4040 GOSUB 590
4050 A$="TERMINE LA TUA MISSIONE."
4060 R=20
4070 GOSUB 600
4080 A$="NEL TEMPO DI"
4090 R=21
4100 GOSUB 600
4110 A$=STR$(T)
4120 C=14
4130 R=21
4140 GOSUB 530
4150 A$="PER GIOCARE ANCORA PREMI S"
4160 R=22
4170 GOSUB 600
4180 FOR KK=1 TO 200
4190 CALL KEY(0,K,S)
4200 IF K=83 THEN 4230
4210 NEXT KK
4220 END
4230 X=0
4240 Y=0
4250 Z=0
4260 VX=0
4270 VY=0
4280 VZ=0
4290 T=0
4300 V=0
4310 PP=0
4320 GOTO 100
4325 REM ***AVARIA***
4330 A$="AVARIA, COMANDI BLOCCATI"
4340 R=23
4350 GOSUB 600
4360 AV=1
4370 GOTO 1020
4380 CALL HCHAR(23,1,32,32)
4390 A$="AVARIA RIPARATA"
4400 R=23
4410 GOSUB 600
4420 AV=0
4430 GOTO 720
```

se, la velocità assoluta e reale è la risultante della somma delle diverse spinte e viene calcolata con una semplice formula matematica: la velocità limite raggiungibile è di 450 pena la distruzione.

Nel corso del gioco noi non conosciamo la posizione degli oggetti presenti nello spazio, ma la navigazione non avviene proprio alla cieca poiché la nave dispone di un radar: premendo il tasto R questo entra in funzione, e in un raggio di 3000 unità segnala la eventuale presenza di corpi celesti e la loro posizione nell'apposita finestra radar dello schermo.

Gli oggetti che possiamo incontrare nello spazio sono stelle, buchi neri, campi di forza, asteroidi nemici e pianeti.

Le stelle e i buchi neri hanno la caratteristica di possedere un campo gravitazionale e quindi di esercitare una forza di attrazione inversamente proporzionale al quadrato della distanza. Il raggio di attrazione delle stelle, (che hanno tutte massa uguale) è di 1500 unità, mentre quello dei buchi neri che esercitano una forza molto maggiore, è di 2000. È molto difficile riuscire a sfuggire alla forza di un buco nero, e se si esita un poco, generalmente si viene distrutti. Per rendere più immediata la comprensione della direzione in cui agisce maggiormente l'attrazione, questa viene scomposta come al solito nei tre vettori paralleli agli assi e il loro valore viene stampato sullo schermo. Questi oggetti hanno anche una superficie propria di raggio 300 (stella) e 500 (B.N.); se la distanza a cui si avvicina la nave è minore, si viene distrutti.

I campi di forza hanno un raggio di 1500 unità, e sono di ostacolo all'avanzamento della nave, in quanto ne riducono continuamente la velocità; una sorta di palude spaziale: uscirne è soprattutto una seccatura.

Le basi nemiche, all'avvicinarsi della nave, ne individuano la rotta e piazzano delle mine; la posizione di queste compare sullo schermo e il giocatore deve evitarle cambiando continuamente traiettoria di volo. Se la mina è evitata si vede solo un bagliore, se si è colpiti si ha un'esplosione in piena regola.

I pianeti infine sono il vero obiettivo della nostra missione. Per scoprire se sono abitati bisogna avvicinarsi ad una distanza inferiore a 500 unità. Così facendo sul video comparirà il messaggio di risposta degli strumenti di rilevamento. Bisogna però porre attenzione a non avvicinarsi ad una distanza minore di 300 unità pena la collisione con il pianeta.

Il programma prevede anche la possibilità di avarie (all'astronave, non al TI 99): i comandi si bloccano e l'astronave diviene ingovernabile per un certo numero di unità di tempo. Il gioco può terminare per completamento della missione, dopo aver trovato un certo numero di pianeti abitati, o in circostanze meno felici, in seguito alla distruzione dell'astronave.

## Analisi del listato

130-180: definizione delle tre matrici e caricamento delle coordinate degli oggetti spaziali, fornite dalla funzione RND.

190: definisce la formula della distanza tra due punti dello spazio.

270-340: queste linee sono il nucleo centrale del programma e definiscono il susseguirsi delle varie routine fondamentali.

350-510: creazione della maschera di stampa cioè di quelle parole o simboli che restano costanti sul video durante il gioco.

530-570: routine per stampare numeri.

580-660: routine per stampare stringhe.

680-940: guida della nave; le righe 700-720 inibiscono la guida in caso di avaria e determinano quando questa ha fine.

940-990: calcolo della nuova posizione della nave nello spazio.

1000-1160: calcolo delle distanze, verifica se siamo troppo vicini a qualche corpo celeste (e quindi distrutti), o se siamo nel suo raggio d'azione (con rinvio alla relativa routine).

1170-1300: cancella dallo schermo i messaggi non più validi.

1300-1700: aggiorna e stampa il tempo e la posizione della nave, la sua velocità e le velocità assiali (queste ultime solo se sono variate).

1710-2220: routine del radar: calcola se c'è qualche oggetto nel raggio di 3000 unità intorno alla nave; stampa le coordinate e il tipo di oggetto.

2230-2420: stampa sul video le coordinate di un corpo celeste col quale siamo in contatto e la distanza della nave da esso.

2430-2560: stampa le forze di attrazione quando esistono.

2570-2670: siamo vicini a una stella.

2680-2830: calcolo della forza di attrazione in relazione alla distanza, scomposizione di questa nei componenti assiali e determinazione della nuova velocità.

2840-2940: siamo vicini a un buco nero.

2950-3330: siamo vicini a una base nemica; la routine calcola dove si troverà la nave dopo due unità di tempo e posiziona una mina in quel punto, ne stampa le coordinate, controlla se gli passiamo vicino e ci segnala se ci siamo salvati.

3340-3450: campo di forza; riduce la velocità della nave ad un terzo di quella precedente.

3450-3860: siamo vicini ad un pianeta; stampa i messaggi di segnalazione, e se siamo a giusta distanza stabilisce casualmente se il pianeta è abitato o no, dà poi nuove coordinate al pianeta per evitare che un giocatore, barando, continui a rivisitarlo.

3870-4320: fine del gioco, o per distruzione o con esito positivo: stampa i vari messaggi, azzerà le variabili, e invita a giocare ancora.

4340-4450: messaggi relativi alla condizione di avaria.

## MEMOR informatica srl

v. Togliatti 4 56030 Perignano Pi

**DISTRIBUISCE ALL'INGROSSO  
IN TUTTA ITALIA**

**Apple computer originali con  
sconti riservati per categorie  
fino al 35 % del listino originale**

**Compatibili 100 % made in italy.**

**Sconti fino 60% del solito listino  
alcuni esempi:**

unita' centrale 64k	640.000
disk-drive slim	330.000
monitor 12" f.v.	159.000
doppio controller card	66.000
super serial card+cavo	138.000
parallel card standard	66.000
scheda 80 col.+ 64 k	175.000
... e tutte le altre periferiche.	

## SOFTWARE garantito:

contabilità semplificata	240.000
contabilità generale	290.000
gestione magazzino	230.000
fatturazione integrata	210.000
gestione effetti	90.000
aggiornamenti assicurati	
e altri 150 pacchetti in sorgente documentati e facili da usare	

**sconti ulteriori per chi opera nel  
settore e per ordinaz. in gruppo**

**listino completo e dettagliato può  
essere richiesto inviando 3.000  
lire in francobolli oppure ordinando  
almeno un articolo in contrassegno**

**Per dettagli tecnici urgenti:  
TELEFONARE allo 0587 - 616084**

**MATERIALI FORNITI CON  
GARANZIA  
SODDISFATTI O RIMBORSATI**

**I prezzi non comprendono l' i.v.a.  
Apple e' un marchio registrato da  
apple computer inc.**